

2011.12.19

### プリント 11, 浮動小数点数

$x, y$  を 16bit の数 (整数). 組  $(x, y)$  で小数を表現する (このような表現を浮動小数点数とよぶ). 講義で説明したように,  $(x, y)$  は

$$x \times 2^{y-15}$$

を表すと約束する.

たとえば (B を付けたら 2 進数)

$x = 0100000000000001B, y = 0$  なら,

$$x \times 2^{-15} = 0.100000000000001B \quad 2 \text{ 進小数}$$

である. たとえば  $x = 000000000000001B, y = 14$  なら,

$$x \times 2^{-1} = 0.100000000000001B \quad 2 \text{ 進小数}$$

である.

1.  $x = \#6000, y = 0$  の時  $(x, y)$  の表す 2 進小数ならびにそれを 10 進小数に直した数はなにか?

2.  $x = \#0FFF, y = 8$  の時  $(x, y)$  の表す 2 進小数ならびにそれを 10 進小数に直した数はなにか?

3. 問題 2 の数を正規化 (normalize) すると  $(x, y)$  はどうなるか?

4. 次のプログラムは (正規化された) 浮動小数点数 (GR1,GR2) と (GR3,GR4) の和を計算して (GR5,GR6) へ格納するプログラムである. 長いプログラムなので分解して考えよう.

1. サブルーチン FNORM は (GR1,GR2) を正規化する. その仕組みをフローチャートにして説明しなさい. FNORM をテストするための入力データを 3 通り考えなさい.
2. FNORM は レジスタ GR3 を作業用に使う. GR3 を保存するようにするように書き換えなさい.
3. ラベル AAA からはどのような処理をしているか?
4. ラベル BBB の時点での各レジスタの値は?
5. サブルーチン FADD の仕組みをフローチャートにして説明しなさい.
6. レポート課題 2. レポートはキャスルシミュレーターのページ

[http://www.chiba-fjb.ac.jp/fjb\\_labo/casl/](http://www.chiba-fjb.ac.jp/fjb_labo/casl/) にあるシミュレータ等を利用してテストして下さい. 要スクリーンコピー等.

このプログラムを改造して浮動小数点数の足し算/引き算をするサブルーチンを作り, その仕様書 (利用するための説明書) を作成しなさい. なお仕様書の例としてそのサブルーチンを利用するプログラムを適当に作成しなさい. 発展問題: 浮動小数点数を 10 進小数として表示するサブルーチンを作り, 上の同様の問題を考えなさい.

```

AA START
;; GR1 * 2^GR2 + GR3 * 2^GR4 --> GR5 * 2^GR6
;; GR1*2^(GR2-15) ( 0x7000*2^(3-15); by asir )
;; GR5=5400, GR6=0005
LAD GR1,#7001
LAD GR2,4
LAD GR3,#7000
LAD GR4,3
CALL FADD
LD GR1,GR5
LD GR2,GR6
CALL FNORM
RET
FADD LD GR6,GR2
CPA GR2,GR4
JMI AAA
LD GR6,GR2 ;; GR2 >= GR4 桁そろえ
LD GR7,GR2
SUBA GR7,GR4
SRL GR3,0,GR7
JUMP BBB
AAA LD GR6,GR4 ;; GR2 < GR4 桁そろえ
LD GR7,GR4
SUBA GR7,GR2
SRL GR1,0,GR7
BBB LD GR5,GR1
ADDL GR5,GR3
PUSH 0,GR5
AND GR5,#8000 ;; 一番上の桁が 1 なら 0 に.
JZE DDD
POP GR5
SRL GR5,1
ADDA GR6,=1
RET
DDD POP GR5
RET
;; normailze GR1=0.1xxxx, GR2 : exponent
FNORM CPA GR1,=0
JNZ FNORMO
RET
FNORMO LD GR3,GR1
AND GR3,#8000
JZE PPP1
SRL GR1,1
ADDA GR2,=1
RET ;; 1xxxx
PPP1 LD GR3,GR1
AND GR3,#4000
JZE PPP2
RET ;; 01xxxx
PPP2 SLL GR1,1 ;; 00xxxx
SUBA GR2,=1
JUMP FNORMO
END

```